

Optimizing DiffEq Code

Chris Rackauckas

August 22, 2020

In this notebook we will walk through some of the main tools for optimizing your code in order to efficiently solve DifferentialEquations.jl. User-side optimizations are important because, for sufficiently difficult problems, most of the time will be spent inside of your `f` function, the function you are trying to solve. "Efficient" integrators are those that reduce the required number of `f` calls to hit the error tolerance. The main ideas for optimizing your DiffEq code, or any Julia function, are the following:

- Make it non-allocating
- Use StaticArrays for small arrays
- Use broadcast fusion
- Make it type-stable
- Reduce redundant calculations
- Make use of BLAS calls
- Optimize algorithm choice

We'll discuss these strategies in the context of small and large systems. Let's start with small systems.

0.1 Optimizing Small Systems (<100 DEs)

Let's take the classic Lorenz system from before. Let's start by naively writing the system in its out-of-place form:

```
function lorenz(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    [dx,dy,dz]
end
```

```
lorenz (generic function with 1 method)
```

Here, `lorenz` returns an object, `[dx,dy,dz]`, which is created within the body of `lorenz`.

This is a common code pattern from high-level languages like MATLAB, SciPy, or R's `deSolve`. However, the issue with this form is that it allocates a vector, `[dx,dy,dz]`, at each step. Let's benchmark the solution process with this choice of function:

```
using DifferentialEquations, BenchmarkTools
u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 10.83 MiB
  allocs estimate: 101435
  -----
  minimum time:      3.310 ms (0.00% GC)
  median time:       3.371 ms (0.00% GC)
  mean time:         4.140 ms (16.31% GC)
  maximum time:      8.585 ms (48.75% GC)
  -----
  samples:           1208
  evals/sample:      1
```

The BenchmarkTools package's `@benchmark` runs the code multiple times to get an accurate measurement. The minimum time is the time it takes when your OS and other background processes aren't getting in the way. Notice that in this case it takes about 5ms to solve and allocates around 11.11 MiB. However, if we were to use this inside of a real user code we'd see a lot of time spent doing garbage collection (GC) to clean up all of the arrays we made. Even if we turn off saving we have these allocations.

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 9.47 MiB
  allocs estimate: 88655
  -----
  minimum time:      2.908 ms (0.00% GC)
  median time:       2.918 ms (0.00% GC)
  mean time:         3.660 ms (14.93% GC)
  maximum time:      9.532 ms (43.34% GC)
  -----
  samples:           1365
  evals/sample:      1
```

The problem of course is that arrays are created every time our derivative function is called. This function is called multiple times per step and is thus the main source of memory usage. To fix this, we can use the in-place form to *****make our code non-allocating*****:

```
function lorenz!(du,u,p,t)
  du[1] = 10.0*(u[2]-u[1])
  du[2] = u[1]*(28.0-u[3]) - u[2]
  du[3] = u[1]*u[2] - (8/3)*u[3]
end

lorenz! (generic function with 1 method)
```

Here, instead of creating an array each time, we utilized the cache array `du`. When the inplace form is used, `DifferentialEquations.jl` takes a different internal route that minimizes

the internal allocations as well. When we benchmark this function, we will see quite a difference.

```
u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5())

BenchmarkTools.Trial:
  memory estimate:  1.38 MiB
  allocs estimate:  12997
  -----
  minimum time:     797.843 μs (0.00% GC)
  median time:      804.057 μs (0.00% GC)
  mean time:        894.675 μs (9.14% GC)
  maximum time:     5.708 ms (80.92% GC)
  -----
  samples:          5576
  evals/sample:     1

@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate:  5.27 KiB
  allocs estimate:  57
  -----
  minimum time:     365.589 μs (0.00% GC)
  median time:      368.290 μs (0.00% GC)
  mean time:        368.719 μs (0.00% GC)
  maximum time:     398.801 μs (0.00% GC)
  -----
  samples:          10000
  evals/sample:     1
```

There is a 4x time difference just from that change! Notice there are still some allocations and this is due to the construction of the integration cache. But this doesn't scale with the problem size:

```
tspan = (0.0,500.0) # 5x longer than before
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate:  5.27 KiB
  allocs estimate:  57
  -----
  minimum time:     1.836 ms (0.00% GC)
  median time:      1.842 ms (0.00% GC)
  mean time:        1.845 ms (0.00% GC)
  maximum time:     1.947 ms (0.00% GC)
  -----
  samples:          2709
  evals/sample:     1
```

since that's all just setup allocations.

But if the system is small we can optimize even more. Allocations are only expensive if they are "heap allocations". For a more in-depth definition of heap allocations, [there are](#)

a lot of sources online. But a good working definition is that heap allocations are variable-sized slabs of memory which have to be pointed to, and this pointer indirection costs time. Additionally, the heap has to be managed and the garbage controllers has to actively keep track of what's on the heap.

However, there's an alternative to heap allocations, known as stack allocations. The stack is statically-sized (known at compile time) and thus its accesses are quick. Additionally, the exact block of memory is known in advance by the compiler, and thus re-using the memory is cheap. This means that allocating on the stack has essentially no cost!

Arrays have to be heap allocated because their size (and thus the amount of memory they take up) is determined at runtime. But there are structures in Julia which are stack-allocated. `structs` for example are stack-allocated "value-type"s. `Tuples` are a stack-allocated collection. The most useful data structure for `DiffEq` though is the `StaticArray` from the package [StaticArrays.jl](#). These arrays have their length determined at compile-time. They are created using macros attached to normal array expressions, for example:

```
using StaticArrays
A = @SVector [2.0,3.0,5.0]

3-element StaticArrays.SArray{Tuple{3},Float64,1,3} with indices SOneTo(3):
 2.0
 3.0
 5.0
```

Notice that the 3 after `SVector` gives the size of the `SVector`. It cannot be changed. Additionally, `SVectors` are immutable, so we have to create a new `SVector` to change values. But remember, we don't have to worry about allocations because this data structure is stack-allocated. `SArrays` have a lot of extra optimizations as well: they have fast matrix multiplication, fast QR factorizations, etc. which directly make use of the information about the size of the array. Thus, when possible they should be used.

Unfortunately static arrays can only be used for sufficiently small arrays. After a certain size, they are forced to heap allocate after some instructions and their compile time balloons. Thus static arrays shouldn't be used if your system has more than 100 variables. Additionally, only the native Julia algorithms can fully utilize static arrays.

Let's `***optimize lorenz using static arrays***`. Note that in this case, we want to use the out-of-place allocating form, but this time we want to output a static array:

```
function lorenz_static(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    @SVector [dx,dy,dz]
end

lorenz_static (generic function with 1 method)
```

To make the solver internally use static arrays, we simply give it a static array as the initial condition:

```
u0 = @SVector [1.0,0.0,0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz_static,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```

BenchmarkTools.Trial:
  memory estimate: 466.30 KiB
  allocs estimate: 2596
  -----
  minimum time:      336.434 μs (0.00% GC)
  median time:       341.281 μs (0.00% GC)
  mean time:         366.297 μs (4.92% GC)
  maximum time:      3.340 ms (86.80% GC)
  -----
  samples:           10000
  evals/sample:      1

@benchmark solve(prob,Tsit5(),save_everystep=false)

```

```

BenchmarkTools.Trial:
  memory estimate: 3.44 KiB
  allocs estimate: 32
  -----
  minimum time:      233.931 μs (0.00% GC)
  median time:       237.636 μs (0.00% GC)
  mean time:         237.795 μs (0.00% GC)
  maximum time:      254.061 μs (0.00% GC)
  -----
  samples:           10000
  evals/sample:      1

```

And that's pretty much all there is to it. With static arrays you don't have to worry about allocating, so use operations like `*` and don't worry about fusing operations (discussed in the next section). Do "the vectorized code" of R/MATLAB/Python and your code in this case will be fast, or directly use the numbers/values.

Exercise 1 Implement the out-of-place array, in-place array, and out-of-place static array forms for the [Henon-Heiles System](#) and time the results.

0.2 Optimizing Large Systems

0.2.1 Interlude: Managing Allocations with Broadcast Fusion

When your system is sufficiently large, or you have to make use of a non-native Julia algorithm, you have to make use of `Arrays`. In order to use arrays in the most efficient manner, you need to be careful about temporary allocations. Vectorized calculations naturally have plenty of temporary array allocations. This is because a vectorized calculation outputs a vector. Thus:

```

A = rand(1000,1000); B = rand(1000,1000); C = rand(1000,1000)
test(A,B,C) = A + B + C
@benchmark test(A,B,C)

```

```

BenchmarkTools.Trial:
  memory estimate: 7.63 MiB
  allocs estimate: 2
  -----
  minimum time:      1.121 ms (0.00% GC)
  median time:       1.144 ms (0.00% GC)
  mean time:         1.280 ms (10.47% GC)

```

```

maximum time:      2.800 ms (57.62% GC)
-----
samples:           3887
evals/sample:      1

```

That expression `A + B + C` creates 2 arrays. It first creates one for the output of `A + B`, then uses that result array to `+ C` to get the final result. 2 arrays! We don't want that! The first thing to do to fix this is to use broadcast fusion. [Broadcast fusion](#) puts expressions together. For example, instead of doing the `+` operations separately, if we were to add them all at the same time, then we would only have a single array that's created. For example:

```

test2(A,B,C) = map((a,b,c)->a+b+c,A,B,C)
@benchmark test2(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  8
-----
 minimum time:     2.180 ms (0.00% GC)
 median time:     2.201 ms (0.00% GC)
 mean time:       2.330 ms (5.43% GC)
 maximum time:    3.838 ms (41.44% GC)
-----
 samples:         2143
 evals/sample:    1

```

Puts the whole expression into a single function call, and thus only one array is required to store output. This is the same as writing the loop:

```

function test3(A,B,C)
    D = similar(A)
    @inbounds for i in eachindex(A)
        D[i] = A[i] + B[i] + C[i]
    end
    D
end
@benchmark test3(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
 minimum time:     1.117 ms (0.00% GC)
 median time:     1.136 ms (0.00% GC)
 mean time:       1.273 ms (10.54% GC)
 maximum time:    2.795 ms (57.64% GC)
-----
 samples:         3909
 evals/sample:    1

```

However, Julia's broadcast is syntactic sugar for this. If multiple expressions have a `.`, then it will put those vectorized operations together. Thus:

```

test4(A,B,C) = A .+ B .+ C
@benchmark test4(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2

```

```

-----
minimum time:      1.122 ms (0.00% GC)
median time:       1.141 ms (0.00% GC)
mean time:         1.277 ms (10.53% GC)
maximum time:      2.810 ms (57.43% GC)
-----
samples:           3896
evals/sample:      1

```

is a version with only 1 array created (the output). Note that `.s` can be used with function calls as well:

```
sin.(A) .+ sin.(B)
```

```

1000×@*(1000 Array{@Float64,2}:
 0.688359 0.809435 0.908384 1.33975 ...@*( 0.925606 1.35092 1.003691.20452 0.956911
0.558767 1.18229 1.39022 0.700042 0.4760731.36832 1.05918 1.18333 0.351968 0.384212
0.820264 0.7534031.39109 1.28623 0.74622 1.18505 0.938225 0.505887 0.7464451.43601
0.371626 1.03881 0.406032 1.28398 0.294736 1.286431.04398 0.780104 0.802716 0.643481
(*@...@*( 1.61725 0.937896 0.4885951.0862 1.50136 0.169723 0.476978 0.397651 0.165244
0.5840440.97933 0.765532 0.754552 1.15242 0.108021 1.4189 0.7611781.30866 1.33338
0.713089 0.231729 1.22802 1.00841 1.604130.638295 0.968986 0.464836 1.21634 0.200881
1.13296 0.692762(*@:@*( (*@` . .@*(1.03473 1.37346 0.643929 1.05397 0.747891 1.23191
1.435721.29449 1.30191 0.895385 0.967597 0.704285 0.800936 1.212930.777125 1.59006
0.51907 1.12099 0.553066 0.453751 1.411291.18067 0.943536 0.732322 1.04078 0.647728
0.945032 0.8163751.36754 0.953724 0.238106 1.29018 (*@...@*( 1.32832 0.684703
0.8826960.768166 0.584829 1.24172 0.946175 0.974031 1.38434 0.1299231.02796 1.11353
0.793456 1.18113 1.2121 1.12809 1.12150.614692 0.814385 0.991451 1.02746 1.02353 1.2313
1.224650.62684 0.64542 1.17342 1.23733 0.640195 1.09594 1.32985

```

Also, the `@.` macro applies a dot to every operator:

```
test5(A,B,C) = @. A + B + C #only one array allocated
@benchmark test5(A,B,C)
```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
  allocs estimate:  2
-----
minimum time:      1.121 ms (0.00% GC)
median time:       1.144 ms (0.00% GC)
mean time:         1.280 ms (10.55% GC)
maximum time:      2.813 ms (58.19% GC)
-----
samples:           3886
evals/sample:      1

```

Using these tools we can get rid of our intermediate array allocations for many vectorized function calls. But we are still allocating the output array. To get rid of that allocation, we can instead use mutation. Mutating broadcast is done via `.=`. For example, if we pre-allocate the output:

```
D = zeros(1000,1000);
```

```

1000×@*(1000 Array{@Float64,2}:
 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 ...@*( 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0
0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 (*@...@*( 0.0

```

```

0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0
0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0(*@:*(
(*@:*( (*@`´.·.*@ (*@:*(0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0
0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 (*@...@*( 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0
0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0

```

Then we can keep re-using this cache for subsequent calculations. The mutating broadcasting form is:

```

test6!(D,A,B,C) = D .= A .+ B .+ C #only one array allocated
@benchmark test6!(D,A,B,C)

```

```

BenchmarkTools.Trial:
  memory estimate:  0 bytes
  allocs estimate:  0
  -----
  minimum time:     1.040 ms (0.00% GC)
  median time:      1.046 ms (0.00% GC)
  mean time:        1.049 ms (0.00% GC)
  maximum time:     1.313 ms (0.00% GC)
  -----
  samples:          4742
  evals/sample:     1

```

If we use `@.` before the `=`, then it will turn it into `.=`:

```

test7!(D,A,B,C) = @. D = A + B + C #only one array allocated
@benchmark test7!(D,A,B,C)

```

```

BenchmarkTools.Trial:
  memory estimate:  0 bytes
  allocs estimate:  0
  -----
  minimum time:     1.042 ms (0.00% GC)
  median time:      1.051 ms (0.00% GC)
  mean time:        1.053 ms (0.00% GC)
  maximum time:     1.313 ms (0.00% GC)
  -----
  samples:          4725
  evals/sample:     1

```

Notice that in this case, there is no "output", and instead the values inside of `D` are what are changed (like with the `DiffEq` inplace function). Many Julia functions have a mutating form which is denoted with a `!`. For example, the mutating form of the `map` is `map!`:

```

test8!(D,A,B,C) = map!((a,b,c)->a+b+c,D,A,B,C)
@benchmark test8!(D,A,B,C)

```

```

BenchmarkTools.Trial:
  memory estimate:  32 bytes
  allocs estimate:  1
  -----
  minimum time:     2.305 ms (0.00% GC)
  median time:      2.314 ms (0.00% GC)

```



```

mean time:      2.316 ms (0.00% GC)
maximum time:   2.526 ms (0.00% GC)
-----
samples:        2155
evals/sample:   1

```

Some operations require using an alternate mutating form in order to be fast. For example, matrix multiplication via `*` allocates a temporary:

```
@benchmark A*B
```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
minimum time:      6.665 ms (0.00% GC)
median time:       6.754 ms (0.00% GC)
mean time:         6.901 ms (1.98% GC)
maximum time:     11.085 ms (0.00% GC)
-----
samples:           724
evals/sample:      1

```

Instead, we can use the mutating form `mul!` into a cache array to avoid allocating the output:

```

using LinearAlgebra
@benchmark mul!(D,A,B) # same as D = A * B

```

```

BenchmarkTools.Trial:
 memory estimate:  0 bytes
 allocs estimate:  0
-----
minimum time:      6.147 ms (0.00% GC)
median time:       6.190 ms (0.00% GC)
mean time:         6.192 ms (0.00% GC)
maximum time:     6.821 ms (0.00% GC)
-----
samples:           807
evals/sample:      1

```

For repeated calculations this reduced allocation can stop GC cycles and thus lead to more efficient code. Additionally, `***`we can fuse together higher level linear algebra operations using BLAS`***`. The package [SugarBLAS.jl](#) makes it easy to write higher level operations like `alpha*B*A + beta*C` as mutating BLAS calls.

0.2.2 Example Optimization: Gierer-Meinhardt Reaction-Diffusion PDE Discretization

Let's optimize the solution of a Reaction-Diffusion PDE's discretization. In its discretized form, this is the ODE:

$$du = D_1(A_y u + u A_x) + \frac{au^2}{v} + \bar{u} - \alpha u \quad (1)$$

$$dv = D_2(A_y v + v A_x) + au^2 + \beta v \quad (2)$$

where u , v , and A are matrices. Here, we will use the simplified version where A is the tridiagonal stencil $[1, -2, 1]$, i.e. it's the 2D discretization of the Laplacian. The native code would be something along the lines of:

```
# Generate the constants
p = (1.0,1.0,1.0,10.0,0.001,100.0) # a,α,ubar,β,D1,D2
N = 100
Ax = Array{Tridiagonal{Float64,3}}{1.0 for i in 1:N-1, [-2.0 for i in 1:N], [1.0 for i in 1:N-1]]
Ay = copy{Array{Tridiagonal{Float64,3}}}(Ax)
Ax[2,1] = 2.0
Ax[end-1,end] = 2.0
Ay[1,2] = 2.0
Ay[end,end-1] = 2.0

function basic_version!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = r[:, :, 1]
    v = r[:, :, 2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    dr[:, :, 1] = Du .+ a.*u.*u./v .+ ubar .- α*u
    dr[:, :, 2] = Dv .+ a.*u.*u .- β*v
end

a,α,ubar,β,D1,D2 = p
uss = (ubar+β)/α
vss = (a/β)*uss^2
r0 = zeros{Float64,3}(100,100,2)
r0[:, :, 1] .= uss.+0.1.*rand{Float64}()
r0[:, :, 2] .= vss

prob = ODEProblem{basic_version!,r0,(0.0,0.1),p}

ODEProblem with uType Array{Float64,3} and tType Float64. In-place: true
timespan: (0.0, 0.1)
u0: [11.01768094418651 11.008180460258423 ...*( 11.02291279551277 11.043884225081605;
11.016674459418661 11.08848630645149 (*@...@*( 11.010182974459266 11.037663098581735;
(*@...@*( ; 11.086862331990217 11.095882881250985 (*@...@*( 11.082339132052795
11.04152101664291; 11.038392100604094 11.035967789389355 (*@...@*(
11.01025899774220411.045152628298178) [12.100000000000001 12.100000000000001 (*@...@*(
12.100000000000001 12.100000000000001; 12.100000000000001 12.100000000000001 (*@...@*(
12.100000000000001 12.100000000000001; (*@...@*( ; 12.100000000000001 12.100000000000001
(*@...@*( 12.100000000000001 12.100000000000001; 12.100000000000001 12.100000000000001
(*@...@*( 12.10000000000000112.100000000000001]
```

In this version we have encoded our initial condition to be a 3-dimensional array, with $u[:, :, 1]$ being the A part and $u[:, :, 2]$ being the B part.

```
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 186.88 MiB
  allocs estimate: 8552
  -----
  minimum time:      48.646 ms (4.03% GC)
  median time:       50.830 ms (7.69% GC)
  mean time:         50.459 ms (7.07% GC)
  maximum time:      51.067 ms (7.82% GC)
  -----
  samples:           100
```

```
evals/sample:      1
```

While this version isn't very efficient,

We recommend writing the "high-level" code first, and iteratively optimizing it! The first thing that we can do is get rid of the slicing allocations. The operation `r[:, :, 1]` creates a temporary array instead of a "view", i.e. a pointer to the already existing memory. To make it a view, add `@view`. Note that we have to be careful with views because they point to the same memory, and thus changing a view changes the original values:

```
A = rand(4)
@show A
B = @view A[1:3]
B[2] = 2
@show A

A = [0.1946507515830036, 0.7999606270351616, 0.06055337919856707, 0.1115153
0494644746]
A = [0.1946507515830036, 2.0, 0.06055337919856707, 0.11151530494644746]
4-element Array{Float64,1}:
 0.1946507515830036
 2.0
 0.06055337919856707
 0.11151530494644746
```

Notice that changing `B` changed `A`. This is something to be careful of, but at the same time we want to use this since we want to modify the output `dr`. Additionally, the last statement is a purely element-wise operation, and thus we can make use of broadcast fusion there. Let's rewrite `basic_version!` to `***avoid slicing allocations***` and to `***use broadcast fusion***`:

```
function gm2!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = @view r[:, :, 1]
    v = @view r[:, :, 2]
    du = @view dr[:, :, 1]
    dv = @view dr[:, :, 2]
    Du = D1*(A*y*u + u*Ax)
    Dv = D2*(A*y*v + v*Ax)
    @. du = Du + a.*u.*u./v + ubar - α*u
    @. dv = Dv + a.*u.*u - β*v
end
prob = ODEProblem(gm2!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
 memory estimate: 119.55 MiB
  allocs estimate: 7082
  -----
 minimum time:      38.736 ms (5.13% GC)
 median time:      38.921 ms (5.22% GC)
 mean time:        39.407 ms (6.36% GC)
 maximum time:     41.110 ms (9.75% GC)
  -----
 samples:          127
 evals/sample:     1
```

Now, most of the allocations are taking place in $Du = D1*(Ay*u + u*Ax)$ since those operations are vectorized and not mutating. We should instead replace the matrix multiplications with `mul!`. When doing so, we will need to have cache variables to write into. This looks like:

```
Ayu = zeros(N,N)
uAx = zeros(N,N)
Du = zeros(N,N)
Ayv = zeros(N,N)
vAx = zeros(N,N)
Dv = zeros(N,N)
function gm3!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = @view r[:,1]
    v = @view r[:,2]
    du = @view dr[:,1]
    dv = @view dr[:,2]
    mul!(Ayu,Ay,u)
    mul!(uAx,u,Ax)
    mul!(Ayv,Ay,v)
    mul!(vAx,v,Ax)
    @. Du = D1*(Ayu + uAx)
    @. Dv = D2*(Ayv + vAx)
    @. du = Du + a*u*u./v + ubar - α*u
    @. dv = Dv + a*u*u - β*v
end
prob = ODEProblem(gm3!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 29.75 MiB
  allocs estimate: 5318
  -----
  minimum time:      33.896 ms (0.00% GC)
  median time:       34.050 ms (0.00% GC)
  mean time:         34.659 ms (1.73% GC)
  maximum time:      36.268 ms (5.58% GC)
  -----
  samples:          145
  evals/sample:     1
```

But our temporary variables are global variables. We need to either declare the caches as `const` or localize them. We can localize them by adding them to the parameters, `p`. It's easier for the compiler to reason about local variables than global variables. *****Localizing variables helps to ensure type stability*****.

```
p = (1.0,1.0,1.0,10.0,0.001,100.0,Ayu,uAx,Du,Ayv,vAx,Dv) # a,α,ubar,β,D1,D2
function gm4!(dr,r,p,t)
    a,α,ubar,β,D1,D2,Ayu,uAx,Du,Ayv,vAx,Dv = p
    u = @view r[:,1]
    v = @view r[:,2]
    du = @view dr[:,1]
    dv = @view dr[:,2]
    mul!(Ayu,Ay,u)
    mul!(uAx,u,Ax)
    mul!(Ayv,Ay,v)
    mul!(vAx,v,Ax)
    @. Du = D1*(Ayu + uAx)
    @. Dv = D2*(Ayv + vAx)
```

```

    @. du = Du + a*u*u./v + ubar -  $\alpha$ *u
    @. dv = Dv + a*u*u -  $\beta$ *v
end
prob = ODEProblem(gm4!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

BenchmarkTools.Trial:
  memory estimate: 29.66 MiB
  allocs estimate: 1054
  -----
  minimum time:      27.273 ms (0.00% GC)
  median time:       27.665 ms (0.00% GC)
  mean time:         28.200 ms (2.11% GC)
  maximum time:      29.824 ms (6.66% GC)
  -----
  samples:          178
  evals/sample:     1

```

We could then use the BLAS `gemmv` to optimize the matrix multiplications some more, but instead let's devectorize the stencil.

```

p = (1.0,1.0,1.0,10.0,0.001,100.0,N)
function fast_gm!(du,u,p,t)
    a, $\alpha$ ,ubar, $\beta$ ,D1,D2,N = p

    @inbounds for j in 2:N-1, i in 2:N-1
        du[i,j,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
            a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
    end

    @inbounds for j in 2:N-1, i in 2:N-1
        du[i,j,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
            a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
    end

    @inbounds for j in 2:N-1
        i = 1
        du[1,j,1] = D1*(2u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
            a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
    end
    @inbounds for j in 2:N-1
        i = 1
        du[1,j,2] = D2*(2u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
            a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
    end

    @inbounds for j in 2:N-1
        i = N
        du[end,j,1] = D1*(2u[i-1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
            a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
    end
    @inbounds for j in 2:N-1
        i = N
        du[end,j,2] = D2*(2u[i-1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
            a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
    end

    @inbounds for i in 2:N-1
        j = 1
        du[i,1,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
            a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
    end
end

```

```

end
@inbounds for i in 2:N-1
    j = 1
    du[i,1,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end

@inbounds begin
    i = 1; j = 1
    du[1,1,1] = D1*(2u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,1,2] = D2*(2u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = 1; j = N
    du[1,N,1] = D1*(2u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,N,2] = D2*(2u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = 1
    du[N,1,1] = D1*(2u[i-1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[N,1,2] = D2*(2u[i-1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = N
    du[end,end,1] = D1*(2u[i-1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[end,end,2] = D2*(2u[i-1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
end
prob = ODEProblem(fast_gm!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
 memory estimate: 29.62 MiB
 allocs estimate: 467
-----
 minimum time:      8.342 ms (0.00% GC)
 median time:       8.445 ms (0.00% GC)
 mean time:         9.024 ms (6.35% GC)
 maximum time:      10.553 ms (18.34% GC)
-----
 samples:           554
 evals/sample:      1

```

Lastly, we can do other things like multithread the main loops, but these optimizations get

the last 2x-3x out. The main optimizations which apply everywhere are the ones we just performed (though the last one only works if your matrix is a stencil. This is known as a matrix-free implementation of the PDE discretization).

This gets us to about 8x faster than our original MATLAB/SciPy/R vectorized style code!

The last thing to do is then `***optimize our algorithm choice***`. We have been using `Tsit5()` as our test algorithm, but in reality this problem is a stiff PDE discretization and thus one recommendation is to use `CVODE_BDF()`. However, instead of using the default dense Jacobian, we should make use of the sparse Jacobian afforded by the problem. The Jacobian is the matrix $\frac{df_i}{dr_j}$, where r is read by the linear index (i.e. down columns). But since the u variables depend on the v , the band size here is large, and thus this will not do well with a Banded Jacobian solver. Instead, we utilize sparse Jacobian algorithms. `CVODE_BDF` allows us to use a sparse Newton-Krylov solver by setting `linear_solver = :GMRES` (see [the solver documentation](#), and thus we can solve this problem efficiently. Let's see how this scales as we increase the integration time.

```
prob = ODEProblem(fast_gm!,r0,(0.0,10.0),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate:  2.76 GiB
  allocs estimate:  41651
  -----
  minimum time:     1.338 s (40.58% GC)
  median time:      1.582 s (38.02% GC)
  mean time:        1.549 s (41.17% GC)
  maximum time:     1.694 s (48.66% GC)
  -----
  samples:          4
  evals/sample:     1
```

```
using Sundials
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))
```

```
BenchmarkTools.Trial:
  memory estimate:  122.60 MiB
  allocs estimate:  20834
  -----
  minimum time:     829.743 ms (0.24% GC)
  median time:      831.897 ms (0.26% GC)
  mean time:        836.368 ms (0.76% GC)
  maximum time:     858.903 ms (3.20% GC)
  -----
  samples:          6
  evals/sample:     1
```

```
prob = ODEProblem(fast_gm!,r0,(0.0,100.0),p)
# Will go out of memory if we don't turn off `save_everystep`!
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate:  2.90 MiB
  allocs estimate:  77
  -----
  minimum time:     5.981 s (0.00% GC)
  median time:      5.981 s (0.00% GC)
  mean time:        5.981 s (0.00% GC)
```

```

maximum time:      5.981 s (0.00% GC)
-----
samples:          1
evals/sample:     1

```

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))
```

```

BenchmarkTools.Trial:
 memory estimate:  324.61 MiB
 allocs estimate:  57069
-----
 minimum time:     2.328 s (0.00% GC)
 median time:     2.366 s (1.24% GC)
 mean time:       2.387 s (2.21% GC)
 maximum time:     2.469 s (5.21% GC)
-----
 samples:         3
 evals/sample:    1

```

Now let's check the allocation growth.

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)
```

```

BenchmarkTools.Trial:
 memory estimate:  3.62 MiB
 allocs estimate:  47243
-----
 minimum time:     2.296 s (0.00% GC)
 median time:     2.306 s (0.00% GC)
 mean time:       2.303 s (0.00% GC)
 maximum time:     2.307 s (0.00% GC)
-----
 samples:         3
 evals/sample:    1

```

```
prob = ODEProblem(fast_gm!,r0,(0.0,500.0),p)
```

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)
```

```

BenchmarkTools.Trial:
 memory estimate:  4.66 MiB
 allocs estimate:  65309
-----
 minimum time:     3.188 s (0.00% GC)
 median time:     3.189 s (0.00% GC)
 mean time:       3.189 s (0.00% GC)
 maximum time:     3.191 s (0.00% GC)
-----
 samples:         2
 evals/sample:    1

```

Notice that we've eliminated almost all allocations, allowing the code to grow without hitting garbage collection and slowing down.

Why is `CVODE_BDF` doing well? What's happening is that, because the problem is stiff, the number of steps required by the explicit Runge-Kutta method grows rapidly, whereas `CVODE_BDF` is taking large steps. Additionally, the `GMRES` linear solver form is quite an efficient way to solve the implicit system in this case. This is problem-dependent, and in many cases using a Krylov method effectively requires a preconditioner, so you need to play around with testing other algorithms and linear solvers to find out what works best with your problem.

0.3 Conclusion

Julia gives you the tools to optimize the solver "all the way", but you need to make use of it. The main thing to avoid is temporary allocations. For small systems, this is effectively done via static arrays. For large systems, this is done via in-place operations and cache arrays. Either way, the resulting solution can be immensely sped up over vectorized formulations by using these principles.

0.4 Appendix

This tutorial is part of the SciMLTutorials.jl repository, found at: <https://github.com/SciML/SciMLTutorials.jl>. For more information on doing scientific machine learning (SciML) with open source software, check out <https://sciml.ai/>.

To locally run this tutorial, do the following commands:

```
using SciMLTutorials
SciMLTutorials.weave_file("introduction", "03-optimizing_diffeq_code.jmd")
```

Computer Information:

```
Julia Version 1.4.2
Commit 44fa15b150* (2020-05-23 18:35 UTC)
Platform Info:
  OS: Linux (x86_64-pc-linux-gnu)
  CPU: Intel(R) Core(TM) i7-9700K CPU @ 3.60GHz
  WORD_SIZE: 64
  LIBM: libopenlibm
  LLVM: libLLVM-8.0.1 (ORCJIT, skylake)
```

Environment:

```
JULIA_LOAD_PATH = /builds/JuliaGPU/DiffEqTutorials.jl:
JULIA_DEPOT_PATH = /builds/JuliaGPU/DiffEqTutorials.jl/.julia
JULIA_CUDA_MEMORY_LIMIT = 2147483648
JULIA_NUM_THREADS = 8
```

Package Information:

```
Status `~/builds/JuliaGPU/DiffEqTutorials.jl/tutorials/introduction/Project.toml`
[6e4b80f9-dd63-53aa-95a3-0cdb28fa8baf] BenchmarkTools 0.5.0
[0c46a032-eb83-5123-abaf-570d42b7fbaf] DifferentialEquations 6.15.0
[65888b18-ceab-5e60-b2b9-181511a3b968] ParameterizedFunctions 5.4.0
[91a5bcdd-55d7-5caf-9e0b-520d859cae80] Plots 1.6.0
[90137ffa-7385-5640-81b9-e52037218182] StaticArrays 0.12.4
[c3572dad-4567-51f8-b174-8c6c989267f4] Sundials 4.2.6
[37e2e46d-f89d-539d-b4ee-838fcccc9c8e] LinearAlgebra
```